
The ISSP module FAMILY was prepared by the research team of Gender Studies, Institute of 

Sociology, Academy of Sciences, Prague. Survey was carried-out by UNIVERSITAS routinely used 

by the Institute of Sociology for its surveys. 

 

 

ISSP background and optional variables 

ISSP standard background and optional variables were created from questions routinely used in 

surveys in the Czech Republic. Questions were translated and are enclosed in this information.  

sex SEX 

age AGE 

marital status MARITAL 

education I EDU_I 

education II EDU_II 

current employment status RESP_EMP 

working hours WORKHOUR 

occupation RESP_OCC 

supervises others in work SUPERVIS 

private vs public  PRIV_PUB 

selfemployed I SELFE_I 

selfemployed II SELFE_II 

spouse/partner: employment status SPOU_EMP 



spouse /partner: occupation SPOU_OCC 

family income F_INCOME 

respondent's earnings EARNINGS 

number of people in household N_HOUSEH 

household composition HC1_AGE to HC7_AGE 

 HC1_SEX to HC7_SEX 

party affiliation  I PARTY_I 

party affiliation II PARTY_II 

religion RELIGION 

attendance of religious services RELI_ATT 

subjective social class SOCCLASS 

region REGION 

size of community SIZE_COM 

urban /rural URB_RUR 

trade unions T_U_MEMB 



Translation of questions from which background variables were created 

Numbers of questions correspond to their numbers in the national version of the questionare. 

 

34. (r34) Are you 

man 1 

woman 2 

 

35. (r35) Which year are you born? 

Last two digits of the year of birth: 

 

36. (r36) What is your marital status? 

Married 1 

Widowed 2 

Divorced 3 

Separated 5 

Single (never married) 6 

 

38. (r38) What is yout highest level of education? 

Incomplete primary 1 

Primary completed 2 

Vocational training, without secondary school diploma 3 

Secondary, without secondary school diploma 4 

Secondary technival with a secondary school diploma 5 

Secondary general with a secondary school diploma 6 

University incomplete 7 

University degree completed 8 

 



39. (r39) How long did you go to school? Include also incomplete or part-time studies, post-

graduate, distance or evening studies.. 

Instruction: If the respondent is a student of secondary school write 95, if he or she is a student of 

university write 96. 

Years: 

 

42. (r42) What is your current economic position, main source of living? 

Employed full-time 1 

Employed part-time (half and more) 2 

Employed part-time (less than half) 3 

Self-employed, entrepreneur 4 

Freelance 5 

Helping family member 6 

Unemployed 7 

Student in school or in vocational training 8 

Permanently disabled 9 

Retired 10 

Housewife /home duties 11 

On maternity leave, maternity benefit 12 

Others 13 

  

43. (r43) Are you a member of a trade union? 

Yes 1 

No 2 



44. (r44) If you are emloyee, do you work in state or private sector? 

State firm 1 

Public institution (schools, health centers, etc.) 2 

Administration (office, ministry, police, army, etc.) 3 

Firm which belongs to a city or municipality 4 

Cooperative 5 

Private firm - native owner 6 

Private firm - foreign owner 7 

 

45. (r45) If you are self-employed, how many employees do you have? 

Instruction: More than 96 write 96, if he or she has no employees write 00. 

Number of employees: 

   

48. (r48) How many hours on average do you work a week? Include also your second jobs, other 

entrepreneurial activities, etc. 

Instruction:More than 96 write 96, refused 97, don't know 98, no answer 99. If respondent doesn't 

work, write 00. 

Number of hours:   

 

49. (r49) What is your occupation in your principal employment? 

Instruction: The answer write as detailed as possible. For example specification "worker" is 

insufficient, there is necessary an exact definition. If the respondent does not work, ask him/her 

about his/her last occupation. If the respondent is a pensioner, ask him/her about his/her 

occupation before he/she retired. 

Coded in ISCO-68 (four digits), then recoded  to three digit  international code. 



50A. (r50a) Do you have subordinates in your principal emloyment? 

Yes 1 

No 2 

 

51. (r51) What is your partner's current economic position, main source of living? 

(same categories as 42.)  

 

52. (r52) What is your partner's occupation? 

(same instruction as 49.) 

 

55. Where do you live? Write a name of region and size of  

Instruction:This answer is written by interviewer.   

(r55o) Name of region: 

 

(r55v)  

Village with less than 1.000 inhabitants 1 

Village with 1.000 - 2.000 inhabitants 2 

Bigger village (2.000 - 5.000 inhabitants) 3 

Small town (5.000 - 10.000 inhabitants) 4 

Bigger town (10.000 - 50.000 inhabitants) 5 

Big town (50.000 - 100.000 inhabitants) 6 

Big town with more  than 100.000 inhabitants, not Prague 7 

Prague 8 

  

59. (r59) How many people are in your household?  

Instruction: More than 8 write 8, no answer or refused write 9. 

Number: 



60. (r60) How old are they? 

Instruction: Fill age and sex of all members of the household except of the respondent, order all 

members by age (the oldest member goes first).  

 1 2 3 4 5 6 7 8 

Age (years):                                                                              

Sex: man 1 1 1 1 1 1 1 1 

 woman 2 2 2 2 2 2 2 2 

 

61A. (r61a) What is your personal average income per month? Count your net income and income 

from additional employment, income from enterprize, unemployment benefits, maternity benefits, 

allowances, pensions, state compensatory benefit for you, do not count child benefits and state 

compensatory benefit for children. 

  

61C. (r61c) What is average net income of your family per month? Count all incomes. 

 

62A. (r62a) Can you tell us what is your religion? 

Write: 

 

63. (r63) How often do you take part in religious ceremonies, services? 

Once a week 1 

2 or 3 times a month 2 

Once a month 3 

Several times a year 4 

Less frequently 5 

Never 6 

 



64. (r64) In politics there are often used terms "the left-hand side" and "the right-hand side". Where 

do you place yourself? 

Far left 1 

Left 2 

Centre 3 

Right 4 

Far right 5 

 

65. (r65) Can you tell us which political party would you in all probability vote in this time? 

Write name of the party:  

 

66. (r66) Into what social class would you classify yourself? 

Lower class 1 

Working class 2 

Lower middle class 3 

Middle class 4 

Upper middle class 5 

Upper class 6 

 

 

 

 

 

 

 

 

 

 



SPSS job for creating background and optional variables for the international merged data file 

 

 

set width 132 length none eject off more off boxstring '-!+' 

screen off 

    listing 'create_1.lis'. 

* . 

get file=   'fam94qcz.sys'. 

* . 

* take ISSP 1994 MODULE questions from Czech Questionnaire. 

* . 

* all these variables have prefix R. 

* only two discrepancies:. 

* 1 - convert question 9: Czech R9A into R9  . 

*                         Czech R9B into R9A . 

* 2 - ISSP question 23 has two series A to D ; solution: . 

*                      R23AA to R23AD are for respondent . 

*                      R23BA to R23BD are for spouse . 

* . 

modify vars /rename (r9a=r9)(r9b=r9a). 

* dummy command to execute transformation. 

descriptives variables= id. 

* . 

* create ISSP standard background variables . 

* note: no variable names advised in ISSP Archiving Rules. 

* . 

compute SEX      = r34. 

compute AGE      = 94-r35. 

compute MARITAL  = r36. 

compute EDU_I    = r39. 

compute EDU_II   = r38. 

recode  EDU_II          (1=2)(2=3)(3 4=4)(5 6 7=5)(8=6)(9=7) 

                        (else=9). 

compute RESP_EMP = r42. 

recode  RESP_EMP        (1=1)(2=2)(3=3)(4=1)(5=1) 

                        (6=4)(7=5)(8=6)(9=7)(10=9) 

                        (11=7)(12=8)(13=1)(14=10) 

                        (else=99). 



compute WORKHOUR = r48. 

compute RESP_OCC = r49. 

* recode specific professions and convert to 3 digit. 

recode  RESP_OCC (3100 thru 3160=3101) 

                 (3320 thru 3380=3399). 

compute RESP_OCC = trunc (RESP_OCC /10). 

if (r49 eq 9997) RESP_OCC = 9997. 

if (r49 eq 9998) RESP_OCC = 9998. 

if (r49 eq 9999) RESP_OCC = 9999. 

compute SUPERVIS = r50a. 

compute PRIV_PUB = r44. 

recode  PRIV_PUB        (0=0)(1=1)(2=2)(3=1)(4=2) 

                        (5 thru 8=3)(97=9)(98=9) 

                        (else=9). 

if ((r42 eq 4) or (r42 eq 5)) PRIV_PUB = 8. 

compute SELFE_I  = r44. 

recode  SELFE_I         (0=0)(else=9). 

if ((r42 eq 4) or (r42 eq 5)) SELFE_I = 1. 

if ((r42 eq 6)              ) SELFE_I = 2. 

compute SELFE_II = r45. 

if ((r42 ne 4) and (r42 ne 5)) SELFE_II = 99. 

compute T_U_MEMB = r43. 

compute SPOU_EMP = r51. 

recode  SPOU_EMP        (1=1)(2=2)(3=3)(4=1)(5=1) 

                        (6=4)(7=5)(8=6)(9=7)(10=9) 

                        (11=7)(12=8)(13=1)(14=10) 

                  (0=0) (else=99). 

compute SPOU_OCC = r52. 

* recode specific professions and convert to 3 digit. 

recode  SPOU_OCC (3100 thru 3160=3101) 

                 (3320 thru 3380=3399). 

compute SPOU_OCC = trunc (SPOU_OCC /10). 

if (r52 eq 9997) SPOU_OCC = 9997. 

if (r52 eq 9998) SPOU_OCC = 9998. 

if (r52 eq 9999) SPOU_OCC = 9999. 

compute F_INCOME = r61c. 

compute EARNINGS = r61a. 

compute N_HOUSEH = r59. 



compute HC1_AGE  = r60_1V. 

compute HC1_SEX  = r60_1P. 

compute HC2_AGE  = r60_2V. 

compute HC2_SEX  = r60_2P. 

compute HC3_AGE  = r60_3V. 

compute HC3_SEX  = r60_3P. 

compute HC4_AGE  = r60_4V. 

compute HC4_SEX  = r60_4P. 

compute HC5_AGE  = r60_5V. 

compute HC5_SEX  = r60_5P. 

compute HC6_AGE  = r60_6V. 

compute HC6_SEX  = r60_6P. 

compute HC7_AGE  = r60_7V. 

compute HC7_SEX  = r60_7P. 

compute PARTY_I  = r64. 

compute PARTY_II = r65. 

compute RELIGION = r62a. 

recode  RELIGION        (1=90)(2=10)(3=49)(4 6 8=92) 

                        (5=20)(7=54)(9=92.500) 

                        (else=99). 

compute RELI_ATT = r63. 

compute SOCCLASS = r66. 

compute REGION   = r55o. 

recode  REGION          (3100 thru 3199 = 1) 

                        (3200 thru 3699 = 2) 

                        (3700 thru 3899 = 3) 

                        (else=9). 

compute SIZE_COM = r55v. 

compute URB_RUR  = r55v. 

recode  URB_RUR         (1 2 3=1)(4 5=2) 

                        (6 7=3)(8=4) 

                        (else=9). 

* dummy command to execute transformation. 

descriptives variables= id. 

* . 

* reorder variables and keep only required ones. 

modify vars 

 /reorder (id r1a to r8 r9 r9a r10 to r23bd sex to urb_rur) 



 /keep    id r1a to r8 r9 r9a r10 to r23bd sex to urb_rur. 

* . 

* map file by descriptives. 

descriptives variables= all. 

save outfile='fam94cz.sys'/compressed. 

finish. 

 

 

 

SPSS job for creating variable labels and value labels  

 

set width 132 length none eject off more off boxstring '-!+' 

screen off 

    listing 'label_1.lis'. 

* . 

get file=    'fam94cz.sys'. 

* . 

* This job gives variables the necessary labels . 

* Missing values are commented below but not set. 

* . 

* make backup copy first. 

save outfile='fam94cz.bak'/compressed. 

get file=    'fam94cz.bak'. 

* . 

* Variable and value labels for original questionnaire vars. 

* *********************************************************. 

* . 

variable labels R1A       'Working mother as..as not working'. 

variable labels R1B       'Working mother_s child suffers'. 

variable labels R1C       'Working woman_s fam suffers'. 

variable labels R1D       'Job OK but women really want home'. 

variable labels R1E       'Be housewife is as..as fulfilling'. 

variable labels R1F       'Job makes woman independent'. 

variable labels R1G       'Women have to work to support fam'. 

variable labels R2A       'Both sexes should feed budget'. 

variable labels R2B       'Man_s job earn, woman_s homefam'. 

variable labels R2C       'Not good househusband, woman in job'. 

variable labels R2D       'Fam suffers thru workoholic men'. 



variable labels R3A       'Woman work after marr, no child'. 

variable labels R3B       'Woman work, underschool child'. 

variable labels R3C       'Woman work, school child'. 

variable labels R3D       'Woman work, grown child'. 

variable labels R4A       'Marr people happier'. 

variable labels R4B       'Marr financial security'. 

variable labels R4C       'Marr purpose child'. 

variable labels R4D       'Bad marr better than none'. 

variable labels R4E       'Should marr if want child'. 

variable labels R4F       'One parent as..as more(two)'. 

variable labels R4G       'Together no marr OK'. 

variable labels R4H       'Good first live then marr'. 

variable labels R4I       'Divorce best marr problem'. 

variable labels R5        'Ideal number child per fam'. 

variable labels R6A       'Watching child grow best joy'. 

variable labels R6B       'Child interfere freedom'. 

variable labels R6C       'No child empty life'. 

variable labels R6D       'Anycase together with child'. 

variable labels R6E       'Anycase together \no child'. 

variable labels R7        'Child obedient\selfthink'. 

variable labels R8        'Mother work below my 14'. 

variable labels R9        'Mother alive'. 

variable labels R9A       'Visit mother freq'. 

variable labels R10       'Ever divorced'. 

variable labels R11       'Spouse ever divorced'. 

variable labels R12       'Lived w\spouse before marr'. 

variable labels R13       'Ever lived unmarr with'. 

variable labels R14A      'Woman maternity leave'. 

variable labels R14B      'Fam??? child_s benefits if work'. 

* Adequate: "Should children receive their benefits also  

*            when the person(s) caring of them work?" . 

variable labels R14C      'Always obtain legal abortion'. 

variable labels R15       'Sex rel before marr wrong'. 

variable labels R16       'Sex rel teens wrong'. 

variable labels R17       'Marr outer rel wrong'. 

variable labels R18       'Same sex rel wrong'. 

variable labels R19       'Met sex harrasment'. 

variable labels R20       'Both incomes managing'. 



variable labels R21A      'Who does laundry'. 

variable labels R21B      'Who does house repairs'. 

variable labels R21C      'Who does sick care'. 

variable labels R21D      'Who does shop groceries'. 

variable labels R21E      'Who does menu decision'. 

variable labels R22       'Who earns more money'. 

variable labels R23AA     'Resp worked after marr, no child'. 

variable labels R23AB     'Resp worked, underschool child'. 

variable labels R23AC     'Resp worked, school child'. 

variable labels R23AD     'Resp worked, grown child'. 

variable labels R23BA     'Spouse worked after marr, no child'. 

variable labels R23BB     'Spouse worked, underschool child'. 

variable labels R23BC     'Spouse worked, school child'. 

variable labels R23BD     'Spouse worked, grown child'. 

* . 

value labels R1A to R2D R4A to R4I R6A to R6E R14A to R14C 

   1 'Strongly agree' 

   2 'Agree' 

   3 'Nei nor' 

   4 'Disagree' 

   5 'Strongly disagree'. 

value labels R3A to R3D R23AA to R23BD  

   1 'Fulltime' 2 'Parttime' 3 'Stayhome'. 

value labels R7  

   1 'Obedient' 2 'Think   self'. 

value labels R8  

   1 'Yes' 2 'No' 3 'Did not livewith'. 

value labels R9  

   1 'Yes' 2 'No'. 

value labels R9A  

   1 'Living  together' 2 'Daily' 3 '+++\week' 4 '1\week'  

   5 '1\month' 6 '+++\year' 7 'Less'. 

value labels R10 R11 R12 

   1 'Yes' 2 'No' 3 'Never   married'. 

value labels R13  

   1 'Yes, w\ previous' 2 'Yes, w\ present' 3 'Yes, w\ both' 

   4 'No'. 

value labels R15 to R18  



   1 'Always  wrong' 2 'Alm alw wrong' 3 'Sometime wrong'  

   4 'No wrong'. 

value labels R19 

   1 'Yes' 2 'No' 3 'Never   worked'. 

value labels R20 

   1 'I all,  p. share' 2 'P. all, I share' 

   3 'Pool all' 4 'Poolpart' 5 'Separate' 6 'Not marr'. 

value labels R21A to R21E 

   1 'Woman+++' 2 'Woman+' 3 'Equal' 4 'Man+' 5 'Man+++'  

   6 'Elseone'. 

value labels R22 

   5 'Woman+++' 4 'Woman+' 3 'Equal' 2 'Man+' 1 'Man+++'. 

* . 

* Missing values survey non-descriptive vars. 

* ******************************************. 

* . 

* missing values labels 

*   R1A to R4I R6A to R9 R10 to R23BD 

*   0 'INAP'    7 'Refused'    8 'Dontknow'    9 'Noanswer'. 

* missing values labels 

*   R5 R9A 

*   0 'INAP'   97 'Refused'   98 'Dontknow'   99 'Noanswer'. 

* . 

* Variable and value labels for ISSP descriptive vars. 

* ***************************************************. 

* . 

*variable labels SEX       ''. 

value labels sex 1 'Male' 2 'Female'. 

*variable labels AGE       ''. 

*variable labels MARITAL   ''. 

value labels MARITAL    

   1 'Married' 2 'Widowed' 3 'Divorced' 4 'Separate' 

   5 'Single'. 

variable labels EDU_I     'Years of schooling'. 

variable labels EDU_II    'Education degree'. 

value labels EDU_II     

   1 'None' 2 'Incompl.primary' 3 'Primary' 4 'Incompl.second.' 

   5 'Second.' 6 'Incompl.Univers.' 7 'Univers.'. 



variable labels RESP_EMP  'Resp_s current employment status'. 

value labels RESP_EMP   

   1 'Fulltime' 2 'Parttime' 3 'Less 15h' 4 'Helping' 

   5 'Unempl. ' 6 'Student ' 7 'Retired ' 8 'Housewif' 

   9 'Disabled' 10 'Others'. 

variable labels WORKHOUR  'Hours worked weekly'. 

variable labels RESP_OCC  'Three digit ISCO\ILO'. 

variable labels SUPERVIS  'Supervise other people'. 

value labels SUPERVIS   

   1 'Yes' 2 'No'. 

variable labels PRIV_PUB  'Work for private\public sector'. 

value labels PRIV_PUB   

   1 'Governm.' 2 'Public' 3 'N.g.n.p.'. 

variable labels SELFE_I   'Self employed or w.for sb.else'. 

value labels SELFE_I    

   1 'Selfemp' 2 'Depend.'. 

variable labels SELFE_II  'Number of employees'. 

variable labels T_U_MEMB  'Trade Union membership'. 

value labels T_U_MEMB   

   1 'Yes' 2 'No'. 

variable labels SPOU_EMP  'Spouse_s current employment status'. 

value labels SPOU_EMP   

   1 'Fulltime' 2 'Parttime' 3 'Less 15h' 4 'Helping' 

   5 'Unempl. ' 6 'Student ' 7 'Retired ' 8 'Housewif' 

   9 'Disabled' 10 'Others'. 

variable labels SPOU_OCC  'Three digit ISCO\ILO'. 

variable labels F_INCOME  'Fam income \nett monthly'. 

variable labels EARNINGS  'Respondent_s earnings \nett monthly'. 

variable labels N_HOUSEH  'Number of people in household'. 

*variable labels HC1_AGE   ''. 

*variable labels HC1_SEX   ''. 

*variable labels HC2_AGE   ''. 

*variable labels HC2_SEX   ''. 

*variable labels HC3_AGE   ''. 

*variable labels HC3_SEX   ''. 

*variable labels HC4_AGE   ''. 

*variable labels HC4_SEX   ''. 

*variable labels HC5_AGE   ''. 



*variable labels HC5_SEX   ''. 

*variable labels HC6_AGE   ''. 

*variable labels HC6_SEX   ''. 

*variable labels HC7_AGE   ''. 

*variable labels HC7_SEX   ''. 

value labels 

  HC1_SEX HC2_SEX HC3_SEX HC4_SEX HC5_SEX HC6_SEX HC7_SEX  

      1 'Male' 2 'Female'. 

variable labels PARTY_I   'Left-right scale'. 

value labels PARTY_I    

   1 'Far left' 2 'Left' 3 'Center' 4 'Right' 5 'Farright'. 

variable labels PARTY_II  'Country specific parties'. 

value labels PARTY_II   

       1 'CSSD    Soc.dem.' 

       2 'KDU-CSL Chr.peo.' 

       3 'CSS     Soc_list' 

       4 'CSZ     Farmers' 

       5 'DSP     Dem.work' 

       6 'DUCHODCI-Retired' 

       7 'HSD-SMS Moravian' 

       8 'KDS     Chr.dem.' 

       9 'KAN     Engag.NP' 

      10 'KSCM    Commies' 

      11 'ODS/KDS Coalit.' 

      12 'LB      Left blk' 

      13 'LSU     LSU' 

      14 'ODA     C.D.All.' 

      15 'ODS     C.D.Par.' 

      16 'OH      C.Movem.' 

      17 'ROI     Gypsies' 

      18 'SPR-RSC Republ.' 

      19 'PODNIK. Entrepr.' 

      20 'ZELENI  Green' 

      40 'JINE    Others' . 

* . 

* Full Party Names in Czech and English / diacritics omitted. 

*   1 'CSSD    Ceskoslovenska socialni demokracie' 

*   1 'CSSD    Czechoslovak Social Democracy' 



*   2 'KDU-CSL Krestansko-demokraticka unie - Ceskoslovenska 

*              strana lidova' 

*   2 'KDU-CSL Christian-Democratic Union - Czechoslovak  

*              People Party' 

*   3 'CSS     Ceskoslovenska strana socialisticka' 

*   3 'CSS     Czechoslovak Socialist Party' 

*   4 'CSZ     Ceskoslovenska strana zemedelska' 

*   4 'CSZ     Czechoslovak Agricultural Party' 

*   5 'DSP     Demokraticka strana prace' 

*   5 'DSP     Democratic Labour Party' 

*   6 'DUCHODCI-Hnuti duchodcu za zivotni jistoty' 

*   6 'DUCHODCI-Movement of Retired for Life Certainty' 

*   7 'HSD-SMS Hnuti za samospravnou demokracii - Spolecnost 

*              pro Moravu a Slezsko' 

*   7 'HSD-SMS Movement for Self-government Democracy -  

*              Association for Moravia and Silesia' 

*   8 'KDS     Krestansko-demokraticka strana' 

*   8 'KDS     Christian Democratic Party' 

*   9 'KAN     Klub angazovanych nestraniku' 

*   9 'KAN     Engaged Non-party-members Club' 

*  10 'KSCM    Komunisticka strana Cech a Moravy' 

*  10 'KSCM    Communist Party of Bohemia and Moravia' 

*  11 'ODS/KDS Koalice ODS/KDS' 

*  11 'ODS/KDS Coalition of ODS and KDS' 

*  12 'LB      Levy blok' 

*  12 'LB      Left block' 

*  13 'LSU     Liberalne socialni unie' 

*  13 'LSU     Liberal Social Union' 

*  14 'ODA     Obcanska demokraticka aliance' 

*  14 'ODA     Citizen Democratic Alliance' 

*  15 'ODS     Obcanska demokraticka strana' 

*  15 'ODS     Citizen Democratic Party' 

*  16 'OH      Obcanske hnuti' 

*  16 'OH      Citizen Movement' 

*  17 'ROI     Romska obcanska iniciativa' 

*  17 'ROI     Gypsy Citizen Initiative' 

*  18 'SPR-RSC Sdruzeni pro republiku - Republikanska strana  

*              Ceskoslovenska' 



*  18 'SPR-RSC Association for Republic - Republican Party  

*              of Czechoslovakia' 

*  19 'PODNIK. Strana ceskych podnikatelu, zivnostniku a rolniku' 

*  19 'PODNIK. Party of Czech Entrepreneurs, Tradesmen and  

*              Farmers' 

*  20 'ZELENI  Strana zelenych' 

*  20 'ZELENI  Party of Greens' 

* . 

*variable labels RELIGION  ''. 

value labels RELIGION   

   10 'Catholic' 20 'Jewish' 49 'Protest.' 54 'Orthodox' 

   90 'None' 92 'Other'. 

variable labels RELI_ATT  'Attendance of religious services'. 

value labels RELI_ATT   

   1 '1\week' 2 '2-3\mnth' 3 '1\month' 4 'X\year' 5 'Less' 

   6 'Never'. 

variable labels SOCCLASS  'Subjective social class'. 

value labels SOCCLASS   

   1 'Lower' 2 'Working' 3 'U.w.L.m.' 4 'Middle' 

   5 'Upp.mid.' 6 'Upper'. 

*variable labels REGION    ''. 

value labels REGION     

   1 'Prague' 2 'Bohemia' 3 'Moravia'. 

variable labels SIZE_COM  ''. 

value labels SIZE_COM   

   1 '    500' 2 '  1,500' 3 '  3,500' 4 '  7,500' 

   5 ' 30,000' 6 ' 75,000' 7 '175,000' 8 '1,2 mill'. 

variable labels URB_RUR   ''. 

value labels URB_RUR    

   1 'Small' 2 'Middle' 3 'Big' 4 'Capital'. 

* . 

* Missing values survey descriptive vars. 

* **************************************. 

* . 

* missing values labels 

*   SEX      MARITAL  SUPERVIS PRIV_PUB SELFE_I  T_U_MEMB  

*   HC1_SEX  HC2_SEX  HC3_SEX  HC4_SEX  HC5_SEX   

*   HC6_SEX  HC7_SEX   



*   PARTY_I  RELI_ATT SOCCLASS REGION   URB_RUR   

*   0 'INAP'    7 'Refused'    8 'Dontknow'    9 'Noanswer'. 

* missing values labels 

*   AGE      EDU_II   RESP_EMP SPOU_EMP N_HOUSEH  

*   HC1_AGE  HC2_AGE  HC3_AGE  HC4_AGE  HC5_AGE   

*   HC6_AGE  HC7_AGE  PARTY_II RELIGION SIZE_COM  

*   0 'INAP'   97 'Refused'   98 'Dontknow'   99 'Noanswer'. 

* missing values labels 

*   RESP_OCC SPOU_OCC  

*   0 'INAP' 9997 'Refused' 9998 'Dontknow' 9999 'Noanswer'. 

* missing values labels \valid zero 

*   SELFE_II F_INCOME EARNINGS  

*              97 'Refused'   98 'Dontknow'   99 'Noanswer'. 

* missing values labels \specific  

*   EDU_I     

*              95 'Studying second.' 96 'Studying univer.' 

*   0 'INAP'   97 'Refused'   98 'Dontknow'   99 'Noanswer'. 

* missing values labels \valid zero and specific 

*   WORKHOUR  

*              96 '96 and  more' 

*              97 'Refused'   98 'Dontknow'   99 'Noanswer'. 

* . 

* map file using descriptives. 

descriptives variables= all. 

* . 

save outfile='fam94cz.sys'/compressed. 

finish. 

 

 

 

List of files for ISSP 1994 Family Module - Czech Republic 

 

FAM94QCZ.POR   SPSS Portable file - original Czech data 

FAM94CZ .POR   SPSS Portable file - Data in ISSP form 

CREATECZ.LOG   SPSS/PC job creating file in ISSP form 

LABELCZ .LOG   SPSS/PC job adding labels to ISSP form 

 



Notes: 

Files contain data and labels, no missings are set on. 

1024 cases, no weighting. 

Files *.por are SPSS Portable, *.log works with SPSS/PC form. 

File fam94qcz.por contains all data as they are defined by the  

     original Czech questionnaire. 

File fam94cz.por contains: 

     identification of each case variable ID, 

     all data variables based on family module questions, 

     core variables (respondents standard characteristics) 

     in correspodance with the ISSP Archiving Rules. 

The conversion to ISSP form is described and commented  

     within the job createcz.log. 

The country specific variables and missing values treatment  

     are described and commented within the job labelcz.log. 

 

Data were processed and Files were created in: 

Institute of Sociology 

Academy of Sciences of the Czech Republic 

Jilska 1 

11000 Praha 1 

Czech Republic 

Phone/operator: (+42 2) 2422 0979 thru 82 

Fax: (+42 2) 2422 0278 

E-mail: socmail@earn.cvut.cz 

  

 

Apendix:  

ISSP Study description form, questionnaire and cards in Czech language. 

 


